
	Connection Object

The Connection object represents a connection to a database.

1. Declare and allocate the Connection.

Private m_DBConnection As ADODB.Connection

Set m_DBConnection = New ADODB.Connection




	Opening the Database
- Either -

1. Set the ConnectString property and use the Open method.

m_DBConnection.ConnectString = _

    "Provider=Microsoft.Jet.OLEDB.3.51;" & _

    "Persist Security Info=False;" & _

    "Data Source=" & db_name

m_DBConnection.Open
- OR -

2. Or use the Open method's parameters.

m_DBConnection.Open _

    "Provider=Microsoft.Jet.OLEDB.3.51;" & _

    "Persist Security Info=False;" & _

    "Data Source=" & db_name




	Executing Commands
Use the Connection's Execute method to execute SQL commands.

m_DBConnection.Execute _

    "CREATE TABLE Employees(" & _

        "LastName
VARCHAR(40)
NOT NULL," & _

        "MiddleName
VARCHAR(40)," & _

        "FirstName
VARCHAR(40)
NOT NULL," & _

        "EmployeeID
INTEGER
NOT NULL," & _

        "SSN
CHAR(9)
NOT NULL," & _

        "Salary
CURRENCY
NOT NULL " & _

    ")"

m_DBConnection.Execute _

    "DROP TABLE Employees"



	Executing Queries
A query statement returns a Recordset that you can use to view the data selected.

1a. Open the Recordset using its Open method.

Dim rs As ADODB.Recordset

Set rs = New ADODB.Recordset

rs.Open query, m_DBConnection, , , adCmdText

1b. Or use the Connection's Execute method.

Dim rs As ADODB.Recordset

Set rs = m_DBConnection.Execute(query, , adCmdText)

2. Examine the records returned.

Do Until rs.EOF

    Debug.Print rs!Name
' Display the name and address.
    Debug.Print rs!Street

    Debug.Print rs!City, rs!State, rs!Zip

    rs.MoveNext
' Get the next record.
Loop




	Closing the Database
It is good practice to close the database connection and set it equal to Nothing when you are through with it.

m_DBConnection.Close

Set m_DBConnection = Nothing




	Recordset Object
A Recordset represents data returned by a query.

Important Methods:

Open
MoveFirst
Delete
GetRows

Close
MoveLast 
AddNew


MoveNext
Update


MovePrevious
CancelUpdate

Important Properties:

BOF
EOF

If BOF and EOF are both True, the Recordset is empty.




	The Fields Collection
The Recordset object's Fields collection contains a Field object for each of the currently selected Recordset's fields.

You can access values by field name, collection key, or collection index.

rs!Name

rs.Fields("Name")

rs.Fields(0)
' Takes about 85% as long.
Important Field Object Properties:

Value
Name
Type




	Editing Records
To modify a record, change its field values and call the Recordset's Update method.

' Find the record to modify.

…

' Make the changes.

rs!Name = new_name

rs!Street = new_street

…

' Save the changes.

rs.Update

To make a new record, use AddNew.

' Make the new record.

rs.AddNew

' Set the field values.

rs!Name = new_name

rs!Street = new_street

…

' Save the changes.

rs.Update




	Tools
This is a collection of tools I have found useful for dealing with databases.

Database code is likely to change in future versions of Visual Basic so I keep all database code in a separate module.

Returning data to the main program may not always be most efficient. For example, if the program must examine and modify every record in a table, it will be faster to process the records using the Recordset directly.




	One Column Queries
These queries are particularly useful for getting a list of values form a dictionary table. You can use the results to initialize ComboBoxes, ListBoxes, etc.

Public Function ExecuteOneColumnQuery( _

    ByVal query As String) As Collection

Dim results As Collection

Dim rs As ADODB.Recordset

    Set results = New Collection

    ' Execute the query.

    Set rs = m_DBConnection.Execute(query, , adCmdText)

    ' Save the results in the collection.

    Do Until rs.EOF

        results.Add rs.Fields(0).Value

        rs.MoveNext

    Loop

    ' Close the Recordset.

    rs.Close

    ' Return the collection.

    Set ExecuteOneColumnQuery = results

End Function




	ComboBox Example
Using the ExecuteOneColumnQuery function, it is easy to initialize a ComboBox using the results returned by a query.
Public Sub InitializeComboFromQuery( _

    ByVal cbo As ComboBox, _

    ByVal query As String, _

    Optional ByVal allow_blank As Boolean = False)

Dim results As Collection

Dim i As Integer

    ' Execute the query.

    Set results = ExecuteOneColumnQuery(query)

    ' If we should allow blanks, start with a blank.

    cbo.Clear

    If allow_blank Then cbo.AddItem ""

    ' Initialize the ComboBox.

    For i = 1 To results.Count

        cbo.AddItem results(i)

    Next i

End Sub




	One Row Queries
This tool packages a row into a Collection instead of a Field objects. Each entry uses its field's name as a key.

Public Function ExecuteOneRowQuery( _

    ByVal query As String) As Collection

Dim results As Collection

Dim rs As ADODB.Recordset

Dim field_num As Integer

    Set results = New Collection

    ' Execute the query.

    Set rs = m_DBConnection.Execute(query, , adCmdText)

    ' Save the results in the collection.

    If Not rs.EOF Then

        For field_num = 0 To rs.Fields.Count - 1

            results.Add rs.Fields(field_num).Value, _

                        rs.Fields(field_num).Name

        Next field_num

    End If

    ' Close the Recordset.

    rs.Close

    ' Return the collection.

    Set ExecuteOneRowQuery = results

End Function




	Multi-Column Queries
The Recordset object's GetRows method copies data from the Recordset into a Variant array.

Public Function ExecuteMultiColumnQuery( _

    ByVal query As String) As Variant

Dim rs As ADODB.Recordset

    ' Execute the query.

    Set rs = m_DBConnection.Execute(query, , adCmdText)

    ' Return the results.

    ExecuteMultiColumnQuery = rs.GetRows()

    ' Close the Recordset.

    rs.Close

End Function




	Getting Field Names
You can use the Recordset object and its Fields collection to get the names of the columns returned by a query.

Public Function GetQueryFields(ByVal query As String) _

    As Collection

Dim results As Collection

Dim rs As ADODB.Recordset

Dim field_num As Integer

    Set results = New Collection

    ' Execute the query.

    Set rs = m_DBConnection.Execute(query, , adCmdText)

    ' Save the field names in the collection.

    For field_num = 0 To rs.Fields.Count - 1

        results.Add rs.Fields(field_num).Name

    Next field_num

    ' Close the Recordset.

    rs.Close

    ' Return the collection.

    Set GetQueryFields = results

End Function




	ComboBox Example, 2
Using the GetQueryFields function, it is easy to initialize a ComboBox using the results returned by a query.
Public Sub InitializeComboFromQueryFields( _

    ByVal cbo As ComboBox, _

    ByVal query As String, _

    Optional ByVal allow_blank As Boolean = False)

Dim results As Collection

Dim i As Integer

    ' Execute the query.

    Set results = GetQueryFields(query)

    ' If we should allow blanks, start with a blank.

    cbo.Clear

    If allow_blank Then cbo.AddItem ""

    ' Initialize the ComboBox.

    For i = 1 To results.Count

        cbo.AddItem results(i)

    Next i

End Sub




	A Useful Example
A query statement returns a Recordset that you can use to view the data selected.

FieldNames is initialize using GetQueryFields. SelectedFields is initially True for all fields.

' A list of the available fields.

Public FieldNames As Collection

' True for selected fields. False for the others.

Public SelectedFields As Collection




	The SELECT Clause
' Build a select clause for the selected fields.

Private Function SelectClause() As String

Dim i As Integer

Dim select_clause As String

    For i = 1 To FieldNames.Count

        If SelectedFields(i) Then

            select_clause = select_clause & _

                FieldNames(i) & ", "

        End If

    Next i

    ' Remove the trailing comma if necessary.

    If Len(select_clause) > 0 Then

        select_clause = "SELECT " & _

            Left$(select_clause, Len(select_clause) - 2)

    Else

        select_clause = "SELECT *"

    End If

    SelectClause = select_clause

End Function




	The WHERE Clause
' Build a WHERE clause from the user's selections.

Private Function WhereClause() As String

Dim where_clause As String

Dim i As Integer

    For i = cboField.LBound To cboField.UBound

        ' See if this row has a non-blank field and operator.

        If Len(cboField(i).Text) > 0 And _

           Len(cboOperator(i).Text) > 0 _

        Then

            ' Add this row to the whare clause.

            If cboOperator(i).Text = "IS NULL" Or _

               cboOperator(i).Text = "IS NOT NULL" _

            Then

                ' IS NULL or IS NOT NULL.

                where_clause = where_clause & _

                    "(" & cboField(i).Text & " " & _

                    cboOperator(i).Text & _

                    ") AND "

            Else

                ' Field Operator Value.

                where_clause = where_clause & _

                    "(" & cboField(i).Text & " " & _

                    cboOperator(i).Text & " '" & _

                    SQLSafe(txtValue(i).Text) & _

                    "') AND "

            End If

        End If

    Next i

    ' Remove the trailing " AND " if necessary.

    If Len(where_clause) > 0 Then

        where_clause = "WHERE " & _

            Left$(where_clause, Len(where_clause) - 5)

    End If

    WhereClause = where_clause

End Function




	The Query
' Make a list of records that match the query.

Private Sub cmdList_Click()

Dim select_clause As String

Dim where_clause As String

Dim query As String

    ' Build the SELECT clause.

    select_clause = SelectClause()

    ' Build the WHERE clause.

    where_clause = WhereClause()

    ' Compose the query.

    query = select_clause & _

        " FROM Customers " & _

        where_clause

    ' Execute the query and display the results in the MSFlexGrid.

    InitializeFlexGridFromQuery Me, flxResults, query

    ' Sort by the selected sort column.

    m_SortColumn = -1

    SortByColumn 0

End Sub




